Compléments

Equations différentielles

On regarde les autres façons de résoudre des équations différentielles.

Solutions développables en séries entières.

```
> restart:
> eq1 := (1-x^2)*diff(y(x),x)-x*y(x)=1:
> Order := 12 : dsolve(eq1,y(x),series);
```

$$\mathrm{y}(x) = \mathrm{y}(0) + x + \frac{1}{2}\,\mathrm{y}(0)\,x^2 + \frac{2}{3}\,x^3 + \frac{3}{8}\,\mathrm{y}(0)\,x^4 + \frac{8}{15}\,x^5 + \frac{5}{16}\,\mathrm{y}(0)\,x^6 + \frac{16}{35}\,x^7 + \frac{35}{128}\,\mathrm{y}(0)\,x^8 + \frac{128}{315}\,x^9 + \frac{63}{256}\,\mathrm{y}(0)\,x^{10} + \frac{256}{693}\,x^{11} + \mathrm{O}(x^{12})$$

```
> dsolve(eq1,y(x));
```

$$\mathrm{y}(x) = \frac{-\ln(x + \sqrt{(-1+x)\,(1+x)}) + \_C1}{\sqrt{-1+x}\,\sqrt{1+x}}$$

Est-il plus facile de developper la solution ?

```
> series(rhs("),x=0,6);
```

$$(-\ln(\%2) + \_C1)\,\%1 - \frac{\%1}{\%2}\,x + (\frac{1}{2}\,(-\ln(\%2) + \_C1)\,\%1 + (\frac{1}{2} + \frac{1}{2}\,\frac{1}{\%2^2})\,\%1)\,x^2 + (-\frac{1}{2}\,\frac{\%1}{\%2} + (-\frac{1}{2}\,\frac{1}{\%2} - \frac{1}{3}\,\frac{1}{\%2^3})\,\%1)\,x^3 +$$
$$(\frac{3}{8}\,(-\ln(\%2) + \_C1)\,\%1 + \frac{1}{2}\,(\frac{1}{2} + \frac{1}{2}\,\frac{1}{\%2^2})\,\%1 + (\frac{1}{4} + \frac{1}{2}\,\frac{1}{\%2^2} + \frac{1}{4}\,\frac{1}{\%2^4})\,\%1)\,x^4 +$$
$$((-\frac{1}{5}\,\frac{1}{\%2^5} - \frac{3}{8}\,\frac{1}{\%2} - \frac{1}{2}\,\frac{1}{\%2^3})\,\%1 - \frac{3}{8}\,\frac{\%1}{\%2} + \frac{1}{2}\,(-\frac{1}{2}\,\frac{1}{\%2} - \frac{1}{3}\,\frac{1}{\%2^3})\,\%1)\,x^5 + \mathrm{O}(x^6)$$
$$\%1 := e^{(1/2\,I\,\mathrm{csgn}(I\,(-1+x))\,\pi)}$$
$$\%2 := e^{(-1/2\,I\,\mathrm{csgn}(I\,(-1+x^2))\,\pi)}$$

Funny spectacular or funny ridiculous ?

```
> yy := (arcsin(x)+a)/sqrt(1-x^2):# ne
> l'appelez pas y !!
> series(yy,x=0);
```

$$a + x + \frac{1}{2}\,a\,x^2 + \frac{2}{3}\,x^3 + \frac{3}{8}\,a\,x^4 + \frac{8}{15}\,x^5 + \frac{5}{16}\,a\,x^6 + \frac{16}{35}\,x^7 + \frac{35}{128}\,a\,x^8 + \frac{128}{315}\,x^9 + \frac{63}{256}\,a\,x^{10} + \frac{256}{693}\,x^{11} + \mathrm{O}(x^{12})$$

Comme quoi il vaut mieux connaître la solution plutôt que de laisser Maple faire à son idée...et chercher une solution développable n'est pas forcément une mauvaise idée ! Vérifions directement

```
> subs(y(x)=yy,eq1);simplify(");
```

$$(1 - x^2)\,(\frac{\partial}{\partial x}\,\frac{\arcsin(x) + a}{\sqrt{1 - x^2}}) - \frac{x\,(\arcsin(x) + a)}{\sqrt{1 - x^2}} = 1$$

$$1 = 1$$

Résolution numérique.

Maple sait faire une résolution purement numérique, avec un large choix de méthodes..., à condition de préciser des conditions initiales !

```
> sol1 :=
> dsolve({eq1,y(0)=1},y(x),numeric,method=classical[foreuler],stepsize
> =0.05);
```

$$sol1 := \mathbf{proc}(x\_classical)\ \ldots\ \mathbf{end}$$

La réponse est une procédure Maple qui à une valeur de x associe une paramétrisation de la courbe intégrale, sous forme d'une équation, ou d'une liste d'équations.

Sans autre indication que numeric, Maple emploie une méthode Runge-Kutta 4-5 avec adaptation du pas.

ASTUCE : On peut extraire une vraie fonction de sol1 en utilisant l'astuce suivante

```
> mysol := t -> subs(sol1(t),y(x)):
```

On a ainsi défini une "vraie" fonction Maple que l'on peut évaluer, résoudre ou tracer.

```
> mysol(sqrt(1/2));
```

$$2.405740958464930$$

```
> fsolve('mysol(t)=2',t=0..0.7);
```

$$.6184129249$$

Les apostrophes sont indispensables pour empêcher Maple d'évaluer prématurément mysol.

Magique non !

Tableaux de résultats.

On peut obtenir les résultats dans un tableau avec l'option value

```
> de1 := {(D@@2)(x)(t)=-y(t),
> (D@@2)(y)(t)=D(x)(t)+y(t)}:
> init1 := {x(0)=1, D(x)(0)=0, y(0)=0, D(y)(0)=1}:
> F := dsolve(de1 union init1, {x(t),y(t)},type=numeric, method=mgear,
> value=array([0,.6,1.1,1.5,2.3,2.5]));
```

$$F := \left[\begin{array}{c} \left[t,\ \mathrm{x}(t),\ \frac{\partial}{\partial t}\,\mathrm{x}(t),\ \mathrm{y}(t),\ \frac{\partial}{\partial t}\,\mathrm{y}(t)\right] \\ \left[\begin{array}{ccccc} 0 & 1. & 0 & 0 & 1. \\ .6 & .9634119549 & -.1848064217 & .6311283580 & 1.148218373 \\ 1.1 & .7669145867 & -.6543597977 & 1.269965676 & 1.421274383 \\ 1.5 & .3877481908 & -1.282704182 & 1.888433444 & 1.670452370 \\ 2.3 & -1.394576262 & -3.372722646 & 3.379308882 & 1.978146378 \\ 2.5 & -2.139341124 & -4.088057357 & 3.773066380 & 1.948716231 \end{array}\right] \end{array}\right]$$

F est une matrice unicolonne dont on va extraire la matrice des résultats.

```
> A:=F[2,1]:evalm(A);
```

$$\left[\begin{array}{ccccc} 0 & 1. & 0 & 0 & 1. \\ .6 & .9634119549 & -.1848064217 & .6311283580 & 1.148218373 \\ 1.1 & .7669145867 & -.6543597977 & 1.269965676 & 1.421274383 \\ 1.5 & .3877481908 & -1.282704182 & 1.888433444 & 1.670452370 \\ 2.3 & -1.394576262 & -3.372722646 & 3.379308882 & 1.978146378 \\ 2.5 & -2.139341124 & -4.088057357 & 3.773066380 & 1.948716231 \end{array}\right]$$

`On peut aussi extraire directement une colonne...`

```
> seq(F[2,1][i,2],i=1..6);
```

$$1.,\ .9634119549,\ .7669145867,\ .3877481908,\ -1.394576262,\ -2.139341124$$

On peut aussi utiliser l'option listprocedure pour avoir directement x(t) en fonction de t.

```
> ff :=
> dsolve({diff(x(t),t)=y(t),diff(y(t),t)=x(t)+y(t),x(0)=2,y(0)=1},
> {x(t),y(t)}, type=numeric, output=listprocedure);
```

$$ff := [t = (\mathbf{proc}(t)\ \ldots\ \mathbf{end}),\ \mathrm{x}(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end}),\ \mathrm{y}(t) = (\mathbf{proc}(t)\ \ldots\ \mathbf{end})]$$

La sortie est une liste de procédures (!) où l'on peut substituer comme précédemment

```
> fx := subs(ff,x(t)): fy := subs(ff,y(t)):
> fx(0), fy(0),fx(12);
```

$$2.,\ 1.,\ .2706691493977095\,10^{9}$$

L'équation de Van der Pol

$y"(x) + (y^2(x) - 1)y'(x) + y(x) = 0$

Dresser le plan de phase, tracer quelques trajectoires, mettre en évidence l'existence d'une trajectoire périodique et montrer que cette trajectoire est un attracteur (ou cycle limite) pour toutes les trajectoires du système.

Divers compléments sur Maple.

Comment récupérer les résultats de solve

```
> mysol := solve({x+y=3,x-y=1});
```

$$mysol := \{x = 2,\ y = 1\}$$

```
> x;x := subs(mysol,x):x;
```

$$x$$

$$2$$

Gradient, laplacien, hessienne

Le package linalg contient tout pour faire des calculs de gradient, laplacien,
matrice hessienne même avec des systèmes de coordonnées non cartésiennes.

```
> restart : with(linalg):

Warning, new definition for norm

Warning, new definition for trace

> f := arctan(y/x):G := grad(f,[x,y]);
```

$$G := \left[-\frac{y}{x^2\,(1+\frac{y^2}{x^2})},\ \frac{1}{x\,(1+\frac{y^2}{x^2})}\right]$$

```
> simplify(G);
```

$$G$$

Cela ne marche pas car G est une liste...donc on utilise map

```
> map(simplify,G);
```

$$\left[-\frac{y}{x^2+y^2},\ \frac{x}{x^2+y^2}\right]$$

```
> laplacian(h(r),[r,theta,phi],coords=spherical
> );
```

$$\frac{2\,r\sin(\theta)\,(\frac{\partial}{\partial r}\,\mathrm{h}(r)) + r^2\sin(\theta)\,(\frac{\partial^2}{\partial r^2}\,\mathrm{h}(r))}{r^2\sin(\theta)}$$

Noter le a dans laplaci<u>a</u>n et dans hessi<u>a</u>n

```
> f := x^3+y^4 + 3*y*z^2:A :=
> hessian(f,[x,y,z]);
```

$$A := \begin{bmatrix} 6\,x & 0 & 0 \\ 0 & 12\,y^2 & 6\,z \\ 0 & 6\,z & 6\,y \end{bmatrix}$$

```
> B := subs(x=1,y=2,z=3,A);
```

$$B := A$$

```
> evalm(B);
```

$$\begin{bmatrix} 6\,x & 0 & 0 \\ 0 & 12\,y^2 & 6\,z \\ 0 & 6\,z & 6\,y \end{bmatrix}$$

Assez surprenant ! en fait on a déja rencontré ce problème, il faut forcer Maple à évaluer la matrice A pour substituer !

```
> B := subs(x=1,y=2,z=3,eval(A)):evalm(B);
```

$$\begin{bmatrix} 6 & 0 & 0 \\ 0 & 48 & 18 \\ 0 & 18 & 12 \end{bmatrix}$$

Pour tester si x est élément d'une liste ou d'un ensemble on utilise la fonction member, avec un paramètre optionnel qui est le rang de la première occurence de x.

```
> l := [1,2,3,7,4,5]:member(4,l,'rg'),rg;
```

$$\mathit{true},\ 5$$

Concaténons où comment créer de vraies fausses suites

Il est impossible en Maple de créer une suite de matrice directement ; l'astuce consiste à utiliser l'opérateur . de concaténation.

```
> A := matrix(2,2,[1,2,3,4]):
> for i to 4 do A.i := evalm(A^i); od:
> A4 : value(A4) : A4[2,2]:k:=4:A.k[2,2]:
```